

FME

CURSO 2019-2020

PROGRAMACIÓN MATEMÁTICA

AMPL

M^aPAZ LINARES, JAVIER HEREDIA, JORDI CASTRO




**Departament d'Estadística
i Investigació Operativa**

UNIVERSITAT POLITÈCNICA DE CATALUNYA

AMPL

- **AMPL: A Modeling Language for Mathematical Programming**
- www.ampl.com

```
model "nombre".mod;  
data "nombre".dat;  
solve;
```




A screenshot of a Windows command window titled "D:\USUARI\MPAZ\DOCENCIA\FME\2012-...". The window contains the following text:

```
ampl: model nombre_modelo.mod;  
ampl: data nombre_datos.dat;  
ampl: solve;_
```

AMPL

- <http://www.ampl.com/IDE/index.html>



- <http://www.neos-server.org/neos/solvers/index.html>

EJEMPLO 1

PLANIFICACIÓN DE LA PRODUCCIÓN

	Cantidades de cada recurso consumidas para producir una unidad del producto A		Cantidades de cada recurso consumidas para producir una unidad del producto B		Disponibilidades totales de cada recurso (por semana)
	Proceso 1	Proceso 2	Proceso 3	Proceso 4	
Personas / Semana	1	1	1	1	15
Kg Material Y	7	5	3	2	120
Kg Material Z	3	5	10	15	100
Beneficio unitario	4	5	9	11	

SOLUCIÓN:

$$\begin{aligned}
 \text{Max} \quad & 4x_1 + 5x_2 + 9x_3 + 11x_4 \\
 & x_1 + x_2 + x_3 + x_4 \leq 15 \\
 & 7x_1 + 5x_2 + 3x_3 + 2x_4 \leq 120 \\
 & 3x_1 + 5x_2 + 10x_3 + 15x_4 \leq 100 \\
 & x_1, x_2, x_3, x_4 \geq 0
 \end{aligned}$$

EJEMPLO 1

PLANIFICACIÓN DE LA PRODUCCIÓN

ejemplo.mod

```
set PRODUCTOS;
set RECURSOS;

param beneficio{PRODUCTOS}>=0;
param recursos_requeridos{PRODUCTOS, RECURSOS} >=0;
param max_recurso{RECURSOS} >=0;

var X {PRODUCTOS}>=0;

# Función objetivo
maximize total:
    sum {i in PRODUCTOS} (beneficio[i]*X[i]);

# Restricciones
# Recursos
subject to recursos_disponibles {j in RECURSOS}:
    sum {i in PRODUCTOS} (recursos_requeridos[i,j]*X[i]) <= max_recurso[j];
```

EJEMPLO 1

PLANIFICACIÓN DE LA PRODUCCIÓN

ejemplo.dat

```
set PRODUCTOS:= 1 2 3 4;
set RECURSOS:= 1 2 3;
param beneficio:=
1 4
2 5
3 9
4 11;

param recursos_requeridos[*,*]
: 1 2 3 :=
1 1 7 3
2 1 5 5
3 1 3 10
4 1 2 15;

param max_recurso:=
1 15
2 120
3 100;
```

EJEMPLO 1

PLANIFICACIÓN DE LA PRODUCCIÓN

ejemplo.run

```
# Borrar los anterior comandos en AMPL
reset;

# Cargar el modelo
model ejemplo.mod;

# Cargar los datos
data ejemplo.dat;

# Indicación a ampl de que se quiere utilizar como solver: CPLEX
option solver cplex;


#Resolver
solve;

# Mostrar los resultados
display total;
display X;
```

EJEMPLO 1

PLANIFICACIÓN DE LA PRODUCCIÓN

include



```
D:\USUARI\MPAZ\DOCENCIA\FME\2012-2013\PROGRAMACIÓ MATEMÀTICA\PROBLEMAS\MOD...
ampl: include ejemplo.run;
CPLEX 11.2.0: optimal solution; objective 99.28571429
3 dual simplex iterations (1 in phase I)
total = 99.2857

X [*] :=
1  7.14286
2  0
3  7.85714
4  0
;

ampl:
```


EJEMPLO 2

FLUJOS EN REDES

Sea:

- $G = (N, A)$ el grafo dirigido definido por un conjunto N de n nodos y un conjunto A de m arcos dirigidos.
- Un vector de producciones o demandas $b \in \mathbb{R}^n$ definido para cada nodo. (producción $\rightarrow +b$, demanda $\rightarrow -b$)
- Un vector de costes $c \in \mathbb{R}^m$ definido para cada arco $c_{ij}, (i, j) \in A$
- Un vector de capacidades $d \in \mathbb{R}^m$ definido para cada arco $d_{ij}, (i, j) \in A$.

Se define la variable x_{ij} como la cantidad de flujo que se envía a través del arco (i, j) .

EJEMPLO 2

FLUJOS EN REDES

FLUJO DE COSTE MÍNIMO

¿De todas las formas posibles de enviar unos determinados flujos entre los nodos de una red, cuál es la de coste total más pequeño?

$$\begin{aligned} \min \quad & \sum_{(i,j) \in A} c_{ij} x_{ij} \\ \text{s. a.} \quad & \sum_{\{j: (i,j) \in A\}} x_{ij} - \sum_{\{j: (j,i) \in A\}} x_{ji} = b_i \quad \forall i \in N \\ & 0 \leq x_{ij} \leq d_{ij} \quad \forall (i,j) \in A \end{aligned}$$

Además el problema debe contar con una red balanceada: $\sum_{i \in N} b_i = 0$



EJEMPLO 2

FLUJOS EN REDES


FLUJO DE COSTE MÍNIMO

Ejemplo:

(c_{ij}, d_{ij})



Coste 555



Coste 535

¿Existe alguna solución con un coste menor? ¿Cuál?

EJEMPLO 2

FLUJOS EN REDES

FLUJO DE COSTE MÍNIMO

flujo_cost_min.mod

```
# Problema de flujo de coste mínimo

#Número de nodos
param n;
set NODOS:=1..n;
set ARCOS within {NODOS,NODOS};
param flujo{NODES};
param coste{ARCOS}>=0;
param capacidad{ARCOS}>=0;
var x {(i,j) in ARCOS}>=0,<=capacidad[i,j];

#Función objetivo
minimize coste_total:
sum{(i,j) in ARCOS} coste[i,j]*x[i,j];

# Restricciones
subject to res_nodos{k in NODOS}:
(sum{(k,j)in ARCOS}x[k,j]-sum{(i,k)in ARCOS} x[i,k])=flujo[k];
```

EJEMPLO 2

FLUJOS EN REDES

FLUJO DE COSTE MÍNIMO

flujo_cost_min.dat

```
# Problema de flujo de coste mínimo
param n:=7;
param flujo:=
7 -45
6 0
5 0
4 0
3 0
2 0
1 45;

param: ARCOS: capacidad coste:=
1 2 20 8
1 3 20 2
1 4 20 4
2 5 15 3
2 7 15 6
3 2 15 7
3 4 15 5
3 5 15 1
3 6 15 5
4 6 5 4
5 6 20 6
5 7 20 9
6 7 25 1;
```

EJEMPLO 2

FLUJOS EN REDES

FLUJO DE COSTE MÍNIMO

flujo_cost_min.run

```
# Borrar los anterior comandos en AMPL
reset;

# Cargar el modelo
model flujo_cost_min.mod;

# Cargar los datos
data flujo_cost_min.dat;

# Indicación a ampl de que se quiere utilizar como solver: CPLEX
#option solver cplex;

#Resolver
solve;

# Mostrar los resultados
display coste_total;
display x;
```

EJEMPLO 2


FLUJOS EN REDES

FLUJO DE COSTE MÍNIMO

¿Existe alguna solución con un coste menor? ¿Cuál?

```
Console
AMPL
cd 'D:\USUARIIS\MPAZ\DOCENCIA\GRADO\2013-2014\OE\2_AMPL\Flujo coste mínimo' ;
ampl: include flujo_cost_min.run;
MINOS 5.5: optimal solution found.
4 iterations, objective 525
coste_total = 525

x :=
1 2 20
1 3 20
1 4 5
2 5 5
2 7 15
3 2 0
3 4 0
3 5 5
3 6 15
4 6 5
5 6 5
5 7 5
6 7 25
;
```



Coste: 525

EJEMPLO 2

FLUJOS EN REDES

FLUJO MÁXIMO

Encuentra el flujo máximo que se puede enviar entre el nodo origen “s” y el nodo destino “t” de una red con capacidades.

$$\max \quad v$$

$$\begin{aligned} \text{s. a.} \quad & \sum_{\{j:(i,j) \in A\}} x_{ij} - \sum_{\{j:(j,i) \in A\}} x_{ji} = 0 \quad \forall i \in N \setminus \{s, t\} \\ & \sum_{\{j:(s,j) \in A\}} x_{sj} = v \\ & \sum_{\{j:(j,t) \in A\}} x_{jt} = v \end{aligned}$$


$$\begin{aligned} x_{ij} &\leq d_{ij} \quad \forall (i, j) \in A \\ x_{ij} &\geq 0 \quad \forall (i, j) \in A, v \geq 0 \end{aligned}$$

EJEMPLO 2

FLUJOS EN REDES

FLUJO MÁXIMO

Ejercicio: Formula con AMPL el siguiente ejemplo y resuélvelo.



EJEMPLO 2

FLUJOS EN REDES

FLUJO MÁXIMO

Solución:

```
Console
AMPL
cd 'D:\USUARIIS\MPAZ\DOCENCIA\GRADO\2013-2014\OE\2_AMPL' ;
ampl: model flujo_maximo.mod;
ampl: data flujo_maximo.dat;
ampl: solve;
MINOS 5.5: optimal solution found.
6 iterations, objective 45
ampl: display flujo_max;
flujo_max = 45

ampl: display flujo;
flujo :=
1 2 20
1 3 20
1 4 5
2 5 15
2 7 15
3 2 10
3 4 0
3 5 5
3 6 5
4 6 5
5 6 0
5 7 20
6 7 10
;
```

